



Programador de Sistemas

Lógica de Programação
Estrutura Sequencial – Exercícios (2ª parte)
Estrutura Condicional - Introdução

23/10/2020

Prof. Mauricio Wieler Orellana


mauricioow@gmail.com

5

● Crie um algoritmo que leia um determinado intervalo de tempo, em dias e converta-o para anos, meses, semanas e dias, sabendo que:

1 ano = 365 dias; 1 mês = 30 dias, 1 semana = 7 dias

Ex: $t = 11119$ dias



```
algoritmo "Conversão de Dias"
```

```
var
```

```
    t, ano, mes, sema, dia: inteiro
```

```
inicio
```

```
    escreval("Digite um intervalo de tempo, em dias: ")
```

```
    leia(t)
```

```
    ano <- t \ 365
```

```
    mes <- (t%365) \ 30
```

```
    sema <- ((t%365)%30)\7
```


```
    dia <- ((t%365)%30)%7
```

```
    escreval(t," dias = ",ano," anos, ",mes," meses, ",sema," semanas  
e ",dia," dias")
```

```
fimalgoritmo
```

②

✿ Crie um algoritmo que leia um saque, em dinheiro e calcule a quantidade de notas de 100, 50, 20, 10, 5, 2 e 1 real necessárias para efetuar esse saque.



Saque de 999 reais → 9 notas de 100, 1 nota de 50,
2 notas de 20, 0 notas de 10, 1 de 5, 2 de 2
0 de 1

Senac

```
algoritmo "Saque"
var
    saque, n100,n50,n20,n10,n5,n2,n1: inteiro
inicio
    escreval("Digite o valor a sacar (em R$): ")
    leia(saque)
    n100 <- saque \ 100
    n50 <- (saque%100)\50
    n20 <- ((saque%100)%50)\20
    n10 <- (((saque%100)%50)%20)\10
    n5 <- (((((saque%100)%50)%20)%10)\5
    n2 <- ((((((saque%100)%50)%20)%10)%5)\2
    n1 <- (((((((saque%100)%50)%20)%10)%5)%2
    escreval("Serão fornecidas: ")
    escreval(n100," notas de R$ 100 ")
    escreval(n50," notas de R$ 50 ")
    escreval(n20," notas de R$ 20 ")
    escreval(n10," notas de R$ 10 ")
    escreval(n5," notas de R$ 5 ")
    escreval(n2," notas de R$ 2 ")
    escreval(n1," notas de R$ 1 ")
fimalgoritmo
```

VB/Delphi: If (condição) Then ^{Definir} End ^{instruções} Else

Java, C#, C:

if (condição verdadeira)

{

→ fin-se

else

{


}

Estrutura Condicional:

Quando queremos efetuar uma decisão (Vou F), usamos a estrutura condicional (de decisão).

Sintaxe: se (condição verdadeira) então
 <instruções>
 senão
 <instruções>
 fin-se

condição falsa



Podemos aninhar os
condicionais :

se <condição1> então
 <instruções>
 senão se <condição2> então
 <instruções>
 fim-se fim-se

Senac

Operadores Lógicos (e/ou/não)

A partir dos valores de verdade verdadeiro, falso, temos:

verdadeiro e verdadeiro = verdadeiro

verdadeiro e falso = falso

falso e verdadeiro = falso

falso e falso = falso

VB/Delphi

E

↓

And

C/C++, C#, Java

E

↓

&&

verdadeiro ou verdadeiro \equiv verdadeiro

verdadeiro ou falso \equiv verdadeiro

falso ou verdadeiro \equiv verdadeiro

falso ou falso \equiv falso

VB/Delphi

OU

↓

Or

C/C++, C#, Java

ou

↓

||

Senac

não:

não verdadeiro \equiv falso

não falso \equiv verdadeiro

\equiv

~~Java, C, C++, C#~~

Op. Relacionais: $>, >=, <, <=, =$

Senac

Exemplo: Crie um algoritmo que verifique se um número é par ou ímpar; inteiro

algoritmo "Par ou Ímpar"

var
x : inteiro

início

escreval("Digite um n° inteiro:")

leia(x)

se $(x \% 2 = 0)$ então

escreval(x, " é par!")

senão

escreval(x, " é ímpar!")

fim
algoritmo

12 | 2
0 | 6

13 | 2
1 | 6

Senac

1. Faça um programa que receba quatro notas de um aluno, calcule e mostre a média aritmética das notas e a mensagem de aprovado ou reprovado, considerando para aprovação média 7.

```
algoritmo "Média de 4 notas"
var
    n1, n2, n3, n4, media: real
    nome: literal
inicio
    escreval("Média de 4 notas do aluno")
    escreval("Digite o nome do aluno: ")
    leia(nome)
    escreval("Digite a 1ª nota do aluno: ")
    leia(n1)
    escreval("Digite a 2ª nota do aluno: ")
    leia(n2)
    escreval("Digite a 3ª nota do aluno: ")
    leia(n3)
    escreval("Digite a 4ª nota do aluno: ")
    leia(n4)
    media <- (n1+n2+n3+n4)/4
    se (media >= 7) entao
        escreval("O aluno ",nome," está APROVADO !!, com média ",media:3:1)
    senao
        escreval("O aluno ",nome," está REPROVADO !!,com média ",media:3:1)
    fimse
fimalgoritmo
```

2. Uma empresa resolve dar um aumento de 12% aos funcionários que recebem um salário inferior a R\$ 2500,00.

Implemente um algoritmo que receba como dado de entrada o salário de um funcionário e imprima o valor do salário reajustado, caso o funcionário tenha direito ao aumento.

Se o funcionário não tiver direito ao aumento, informe isso através de uma mensagem.

```
algoritmo "Aumento de Salário"
var
    sal, novo_sal, aum: real
inicio
    escreval("Digite o salário do funcionário (em R$): ")
    leia(sal)
    se (sal <= 500) entao
        aum <- 0.30*sal //0.30 = 30% = 30/100
        novo_sal <- sal + aum
        escreval("O salário será reajustado em
R$",aum:4:2)
        escreval("O salário com aumento é de
R$",novo_sal:6:2)
    senao
        escreval("O funcionário não tem direito a
aumento !")
    fimse
finalgoritmo
```


3. Em uma determinada disciplina, a nota final do estudante é calculada a partir de 3 notas atribuídas a um trabalho, a um teste em laboratório e a uma prova escrita em sala de aula, respectivamente. As notas variam de 0 a 10 e a nota final é a média ponderada das 3 notas anteriormente citadas. A seguir são apresentados os pesos das notas:

- Trabalho: peso 2
- Teste em laboratório: peso 2
- Prova escrita em sala de aula: peso 6

Sabendo que foram dadas 64 aulas, implemente um programa que receba as 3 notas do estudante e o número de faltas, calcule e imprima a média final e uma mensagem de acordo com as seguintes regras:

- Se o número de faltas for superior a 25% das aulas dadas, REPROVADO POR FALTA
- Se o número de faltas for inferior ou igual a 25% das aulas dadas e a média final for igual ou superior a 7,0: ALUNO APROVADO
- Se o número de faltas for inferior ou igual a 25% das aulas dadas e a média final for inferior a 7,0 mas superior ou igual a 5,0: EXAME
- Se a média final for inferior a 5,0: ALUNO REPROVADO

Para calcular a média, utilize a fórmula:

$$\text{média} = ((\text{Nota do Trabalho} * 2) + (\text{Nota do Teste} * 2) + (\text{Nota da Prova} * 6)) / 10$$

```

algoritmo "Notas de Aluno"
var
    nt, ntl, ntpe, media, nf: real
    nome: literal
inicio
    escreval("NOTAS DE UM ALUNO")
    escreval("Digite o nome do aluno: ")
    leia(nome)
    escreval("Digite a nota do trabalho de ", nome)
    leia(nt)
    escreval("Digite a nota do teste de lab. de ", nome)
    leia(ntl)
    escreval("Digite a nota da prova escrita de ", nome)
    leia(ntpe)
    escreval("Digite o n° de faltas de ", nome)
    leia(nf)
    media <- (nt*2 + ntl*2 + ntpe*6)/10
    se (nf > 16) entao
        escreval(nome, " foi Reprovado por Faltas")
    fimse
    se ((nf <= 16) e (media >= 7)) entao
        escreval(nome, " foi Aprovado com média: ", media:3:1)
    fimse
    se ((nf <= 16) e (media >= 5) e (media < 7)) entao
        escreval(nome, " ficou de exame e sua média foi de: ", media:3:1)
        escreval(nome, " precisa de ", (10-media):3:1, " para aprovar")
    fimse
    se (media < 5) entao
        escreval(nome, " está REPROVADO e teve média: ", media:3:1)
    fimse
fimalgoritmo

```

4.

👤 Construa um algoritmo para determinar se o indivíduo esta com um peso favorável. Essa situação é determinada através do IMC (Índice de Massa Corpórea), que é definida como sendo a relação entre o peso (PESO) e o quadrado da Altura (ALTURA) do indivíduo. Ou seja:

$$IMC = \frac{PESO}{ALTURA^2}$$

e, a situação do peso é determinada pela tabela abaixo:

Condição	Situação
IMC abaixo de 20	Abaixo do peso
IMC de 20 até 25	Peso Normal
IMC de 25 até 30	Sobre Peso
IMC de 30 até 40	Obeso
IMC de 40 e acima	Obeso Mórbito

```
Se (idade >= 18) entao
    escreval("Maior de Idade")
senao se (idade >= 12) escreval("Adolescente")
    senao se (idade >= 3) escreval("Criança")
        senao escreval("Bebê")
```

```
fimse
```

```
Se (idade >= 18) entao
· escreval("Maior de Idade")
fimse
Se (idade <18 e idade >=12)
    escreval("Adolescente")
fimse
Se (idade <12 e idade >=3)
    escreval("Criança")
Fimse
Se (idade < 3 e idade >=0)
    escreval("Bebê")
fimse
```

```
Se (idade >= 18) entao
    escreval("Maior de Idade")
    senao se (idade >= 12) escreval("Adolescente")
        senao se (idade >= 3) escreval("Criança")
            senao escreval("Bebê")
        fimse
    fimse
fimse
```

```

algoritmo "IMC"
var
    peso, altu, imc: real
    nome: literal
inicio
    escreval("I M C Índice de Massa Corpórea")
    escreval("Digite o nome da pessoa: ")
    leia(nome)
    escreval("Digite o peso da pessoa, em kg: ")
    leia(peso)
    escreval("Digite a altura da pessoa, em m: ")
    leia(altu)
    imc <- peso/(altu^2)
    escreval("Situação de ", nome)
    se (imc < 20) entao
        escreval("IMC = ", imc:3:1, " Abaixo do Peso")
    senao
        se ((imc >= 20) e (imc <= 25)) entao
            escreval("IMC = ", imc:3:1, " Peso Normal")
        senao
            se ((imc > 25) e (imc <= 30)) entao
                escreval("IMC = ", imc:3:1, " Sobrepeso")
            senao
                se ((imc > 30) e (imc <= 40)) entao
                    escreval("IMC = ", imc:3:1, " Obeso")
                senao
                    escreval("IMC = ", imc:3:1, " Obeso Mórbido")
            fimse
        fimse
    fimse
fimse
fimalgoritmo

```

5. Escreva um algoritmo para determinar se um número A é divisível por um outro número B. Esses valores devem ser fornecidos pelo usuário. Detalhe: Se B for zero (0), o algoritmo deve retornar uma mensagem, dizendo que não 's possível haver divisão por zero!

Ex: A = 24 B = 8 => A é divisível por B

A = 35 B = 10 => A não é divisível por B

A = 54 B = 0 => não existe divisão por zero

$$\rightarrow 24 \% 8 = 0$$
$$\begin{array}{r} 24 \overline{) 8} \\ \underline{0} \\ 8 \\ \underline{0} \\ 0 \end{array}$$

```
algoritmo "Divisibilidade"
var
    A, B : inteiro
inicio
    escreval("Divisibilidade entre dois números: ")
    escreval("Digite o 1º número: ")
    leia(A)
    escreval("Digite o 2º número: ")
    leia(B)
    se (B = 0) entao
        escreval("Não existe divisão por zero !")
    senao
        se (A % B = 0) entao
            escreval(A, " é divisível por ", B)
        senao
            escreval(A, " não é divisível por ", B)
        fimse
    fimse
fimalgoritmo
```

Senac

6. Construa um algoritmo que imprima qual o menor e qual o maior valor de dois números A e B, lidos através do teclado.



```
algoritmo "Compara dois números"
var
    A, B : inteiro
inicio
    escreval("Comparação entre dois números: ")
    escreval("Digite o 1º número: ")
    leia(A)
    escreval("Digite o 2º número: ")
    leia(B)
    se (A > B) entao
        escreval("O 1º número é maior que o 2º número")
    senao
        se (A = B) entao
            escreval("O 1º número é igual ao 2º
número")
        senao
            escreval("O 1º número é menor que o 2º
número")
        fimse
    fimse
fimalgoritmo
```

7. Escreva um algoritmo que leia um número inteiro e informe se ele é divisível por 3 e por 7, ao mesmo tempo.

```
algoritmo "Divisibilidade por 3 e por 7"
var
    x : inteiro
inicio
    escreval("Divisibilidade por 3 e por 7: ")
    escreval("Digite um número inteiro: ")
    leia(x)
    se (x % 3 = 0) entao
        escreval(x, " é divisível por 3")
    fimse
    se (x % 7 = 0) entao
        escreval(x, " é divisível por 7")
    fimse
    se (x % 3 = 0) e (x % 7 = 0) entao
        escreval(x, " é divisível por 3 e por 7")
    senao
        escreval(x, " NÃO é divisível por 3 e por 7")
    fimse
fimalgoritmo
```

4. A prefeitura de Guarulhos abriu uma linha de crédito para os funcionários estatutários.
O valor máximo da prestação não poderá ultrapassar 30% do salário bruto.
Crie um algoritmo que permita entrar com o salário bruto e o valor da prestação, e informar se o empréstimo pode ou não ser concedido.

Senac

```
algoritmo "Empréstimo casa própria"
var
    sal_bruto,prest : real
inicio
    escreval("Empréstimo para casa própria: ")
    escreval("Digite o salário bruto da pessoa: ")
    leia(sal_bruto)
    escreval("Digite o valor da prestação que ela quer pagar: ")
    leia(prest)
    se (prest <= 0.30*sal_bruto) entao
        escreval("Empréstimo pode ser concedido!")
    senao
        escreval("Empréstimo NÃO pode ser concedido!")
    fimse
fimalgoritmo
```

8. Escreva um algoritmo que leia um número e informe se ele é divisível por 10, por 5 ou por 2 ou se não é divisível por nenhum deles.

Senac


```
algoritmo "Divisibilidade por 10, 5 ou 2"
var
    num : inteiro
inicio
    escreval("Divisibilidade por 10, 5 ou 2")
    escreval("Digite um número inteiro: ")
    leia(num)
    se (num%10 = 0) entao
        escreval(num, " é divisível por 10")
    senao
        escreval(num, " não é divisível por 10")
    fimse
    se (num%5 = 0) entao
        escreval(num, " é divisível por 5")
    senao
        escreval(num, " não é divisível por 5")
    fimse
    se (num%2 = 0) entao
        escreval(num, " é divisível por 2")
    senao
        escreval(num, " não é divisível por 2")
    fimse
finalgoritmo
```

9. Um comerciante calcula o valor da venda, tendo em vista a tabela a seguir:

Valor da Compra	Valor da Venda
Valor < R\$ 10,00	Lucro de 70%
R\$ 10,00 ≤ Valor < R\$ 30,00	Lucro de 50%
R\$ 30,00 ≤ Valor < R\$ 50,00	Lucro de 40%
Valor ≥ R\$ 50,00	Lucro de 30%

Criar um algoritmo que leia o valor da compra e imprima o valor da venda.

Senac

```

algoritmo "Compra venda"
var
    preco, lucro, preco_venda:real
inicio
    escreval("P R E Ç O S")
    escreval("Digite o preço da compra(em R$): ")
    leia(preco)
    se (preco < 10) entao
        lucro <- preco*0.70
    senao
        se ((preco >= 10) e (preco < 30)) entao
            lucro <- preco * 0.50
        senao
            se ((preco >= 30) e (preco < 50)) entao
                lucro <- preco*0.40
            senao
                lucro <- preco*0.30
        fimse
    fimse
    fimse
    preco_venda <- preco+ lucro
    escreval("O lucro será de R$ ",lucro:4:2)
    escreval("Preço de Venda = R$ ",preco_venda:4:2)
fimalgoritmo

```

Senac



Faça um programa que receba o preço de um produto, calcule e mostre, de acordo com as tabelas a seguir, o novo preço e a classificação.

TABELA 1 - PERCENTUAL DE AUMENTO

PREÇO	%
Até R\$ 50,00	5
Entre R\$ 50,00 e R\$ 100,00	10
Acima de R\$ 100,00	15

TABELA 2 - CLASSIFICAÇÕES

NOVO PREÇO	CLASSIFICAÇÃO
Até R\$ 80,00	Barato
Entre R\$ 80,00 e R\$ 120,00 (inclusive)	Normal
Entre R\$ 120,00 e R\$ 200,00 (inclusive)	Caro
Maior que R\$ 200,00	Muito caro

```

algoritmo "Preços com Aumento"
var
    preco_prod, aum, novo_preco:real
inicio
    escreval("P R E Ç O S")
    escreval("Digite o preço do produto (em R$): ")
    leia(preco_prod)
    se (preco_prod <= 50) entao
        aum <- preco_prod*0.05
    senao
        se ((preco_prod > 50) e (preco_prod <=100)) entao
            aum <- preco_prod * 0.10
        senao
            aum <- preco_prod*0.15
        fimse
    fimse
    novo_preco <- preco_prod + aum
    se (novo_preco <= 80) entao
        escreval("Novo Preço = R$ ",novo_preco:4:2," BARATO")
    fimse
    se ((novo_preco > 80) e (novo_preco <= 120)) entao
        escreval("Novo Preço = R$ ",novo_preco:4:2," NORMAL")
    fimse
    se ((novo_preco > 120) e (novo_preco <= 200)) entao
        escreval("Novo Preço = R$ ",novo_preco:4:2," CARO")
    fimse
    se (novo_preco >= 200) entao
        escreval("Novo Preço = R$ ",novo_preco:4:2," MUITO CARO")
    fimse
fimalgoritmo

```

12. Dados três valores, A, B e C, construa um algoritmo para verificar se estes valores podem ser valores dos lados de um triângulo.

A, B e C formam um triângulo $\Leftrightarrow A < (B+C)$ e $B < (A + C)$ e
 $C < (A + B)$

```
algoritmo "Triângulos"
var
    A,B,C: inteiro
inicio
    escreval("Verificação de Existência de Triângulos")
    escreval("Digite o 1º número: ")
    leia(A)
    escreval("Digite o 2º número: ")
    leia(B)
    escreval("Digite o 3º número: ")
    leia(C)
    se (A < (B+C)) e (B < (A+C)) e (C < (A+B)) entao
        escreval(A,",", ",B," e ",C," formam um triângulo ")
    senao
        escreval(A,",", ",B," e ",C," não formam um triângulo ")
    fimse
fimalgoritmo
```


13. Dados três valores, A, B e C, construa um algoritmo para verificar se estes valores podem ser valores dos lados de um triângulo, e se for, se é um triângulo escaleno, um triângulo equilátero ou um triângulo isósceles.

Se $(A = B) \text{ e } (B = C) \Rightarrow$ Triângulo equilátero

Se $(A = B) \text{ ou } (B = C) \text{ ou } (A = C) \Rightarrow$ Triângulo isósceles

Se $(A \neq B) \text{ e } (B \neq C) \text{ e } (A \neq C) \Rightarrow$ Triângulo escaleno

```

algoritmo "Classificação de Triângulos"
var
    A,B,C: inteiro
inicio
    escreval("Verificação de Existência de Triângulos")
    escreval("Digite o 1º número: ")
    leia(A)
    escreval("Digite o 2º número: ")
    leia(B)
    escreval("Digite o 3º número: ")
    leia(C)
    limpatela
    escreval
    se (A < (B+C)) e (B < (A+C)) e (C < (A+B)) entao
        escreval(A," ",B," e ",C," formam um triângulo ")
        se (A = B) e (B = C) entao
            escreval("EQUILÁTERO")
        fimse
        se (A = B) ou (B = C) ou (A=C) entao
            escreval("ISÓSCELES")
        fimse
        se (A <> B) e (B <> C) e (A <> C) entao
            escreval("ESCALENO")
        fimse
    senao
        escreval(A," ",B," e ",C," não formam um triângulo ")
    fimse
fimalgoritmo

```

14. Recomendam-se estudantes para bolsas de estudo em função de seu desempenho anterior. A natureza das recomendações é baseada na seguinte tabela:

Média	Recomendação
$Média \geq 9,0$	Altamente recomendado
$8,0 \leq Média < 9,0$	Fortemente recomendado
$7,0 \leq Média < 8,0$	Recomendado
$Média < 7,0$	Não recomendado

Faça um programa para ler o nome e a média de um aluno e determinar sua recomendação. O programa deverá exibir a seguinte mensagem:

O aluno _____ é _____.

```
algoritmo "Estudantes e bolsas de estudo"

var
    nome:literal
    media :real
inicio
    escreval("Estudantes e bolsas de estudo")
    escreval("Digite o nome do aluno")
    leia(nome)
    escreval("Digite a média do aluno")
    leia(media)
    se (media >= 9) entao
        escreval("O aluno ",nome," é altamente recomendado.")
    fimse
    se (media >= 8) e (media < 9) entao
        escreval("O aluno ",nome," é fortemente recomendado.")
    fimse
    se (media >= 7) e (media < 8) entao
        escreval ("O aluno ", nome," é recomendado.")
    fimse
    se (media < 7) entao
        escreval("O aluno ",nome," não é recomendado.")
    fimse

finalgoritmo
```

Senac

15. Faça um algoritmo que receba três números e mostre-os em ordem crescente. Suponha que o usuário digitará três números diferentes.

Se $(n1 < n2)$ e $(n2 < n3) \Rightarrow$ Ordem Crescente: $n1, n2, n3$

Se $(n1 < n3)$ e $(n3 < n2) \Rightarrow$ Ordem Crescente: $n1, n3, n2$

Se $(n2 < n1)$ e $(n1 < n3) \Rightarrow$ Ordem Crescente: $n2, n1, n3$

Se $(n2 < n3)$ e $(n3 < n1) \Rightarrow$ Ordem Crescente: $n2, n3, n1$

Se $(n3 < n1)$ e $(n1 < n2) \Rightarrow$ Ordem Crescente: $n3, n1, n2$

Se $(n3 < n2)$ e $(n2 < n1) \Rightarrow$ Ordem Crescente: $n3, n2, n1$

```

algoritmo "Classificação de 3 números "
var
    n1,n2,n3: inteiro
inicio
    escreval("Classifica em ordem crescente três números inteiros: ")
    escreval("Digite o 1º número: ")
    leia(n1)
    escreval("Digite o 2º número: ")
    leia(n2)
    escreval("Digite o 3º número: ")
    leia(n3)
    escreval("Ordem Crescente:")
    se ((n1 < n2) e (n2 < n3)) entao
        escreval(n1," ",n2," ",n3)
    fimse
    se ((n1 < n3) e (n3 < n2)) entao
        escreval(n1," ",n3," ",n2)
    fimse
    se ((n2 < n3) e (n3 < n1)) entao
        escreval(n2," ",n3," ",n1)
    fimse
    se ((n2 < n1) e (n1 < n3)) entao
        escreval(n2," ",n1," ",n3)
    fimse
    se ((n3 < n1) e (n1 < n2)) entao
        escreval(n3," ",n1," ",n2)
    fimse
    se ((n3 < n2) e (n2 < n1)) entao
        escreval(n3," ",n2," ",n1)
    fimse
fimalgoritmo

```