



**Programador de Sistemas**

**Lógica de Programação**

**Vetores – 2ª parte**

Prof. Mauricio Wieler Orellana

[mauricioow@gmail.com](mailto:mauricioow@gmail.com)

1. Faça um algoritmo que leia 10 números, armazene-os num vetor, e mostre-os na ordem inversa.

	1	2	3	4	5	6	7	8	9	10	índices
V	5	-1	8	15	13	-7	23	41	-1	0	

Deve imprimir, na tela:

	1	2	3	4	5	6	7	8	9	10
VINV	0	-1	41	23	-7	13	15	8	-1	5
	10	9	8	7	6	5	4	3	2	1

1º.

2º.

```
algoritmo "Vetor Invertido"
var
    VET:vetor[1..10] de inteiro
    i:inteiro
inicio
    //Rotina de leitura
    para i de 1 ate 10 faca
        escreval("Digite VET[,i,]:")
        leia(VET[i])
    fimpara
    limpatela
    escreval("Vetor na ordem de leitura normal:")
    para i de 1 ate 10 faca
        escreva(VET[i]," ")
    fimpara
    escreval
    escreval
    escreval("Vetor na ordem de leitura invertida:")
    para i de 10 ate 1 passo -1 faca
        escreva(VET[i]," ")
    fimpara
    escreval
fimalgoritmo
```

28 Faça um programa que leia um vetor de 15 posições de números inteiros e divida todos os seus elementos pelo maior valor do vetor. Mostre o vetor após os cálculos.



```

algoritmo "Vetor dividido pelo maior"
var
    VET,W:vetor[1..15] de real
    i:inteiro
    maior:real
inicio
    //Rotina de leitura
    para i de 1 ate 15 faca
        escreval("Digite VET[",i,"]:")
        leia(VET[i])
    fimpara
    maior <- VET[1]
    para i de 2 ate 15 faca
        se (VET[i] > maior) entao
            maior <- VET[i]
        fimse
    fimpara
    limpatela
    escreval("Vetor digitado pelo usuário:")
    para i de 1 ate 15 faca
        escreval(VET[i])
        W[i] <- VET[i]/maior
    fimpara
    escreval
    escreval
    escreval("Maior número digitado: ",maior)
    escreval("Vetor dividido pelo maior número")
    para i de 1 ate 15 faca
        escreval(W[i]:3:2)
    fimpara
    escreval
fimalgoritmo

```

1. Faça um programa que receba o nome de cinco produtos e seus respectivos preços, calcule e mostre:
- a quantidade de produtos com preço inferior a R\$ 50,00;
  - o nome dos produtos com preço entre R\$ 50,00 e R\$ 100,00;
  - a média dos preços dos produtos com preço superior a R\$ 100,00.



```

algoritmo "Nomes e preços em vetores"
var
    NOME:vetor[1..5] de caracter
    PRECO:vetor[1..5] de real
    medial00,somal00:real
    i,cont50,cont100: inteiro
inicio
    //Rotina de leitura
    cont50 <- 0
    cont100 <- 0
    somal00 <- 0
    escreval("Carregando os vetores NOME e PRECO com 5 elementos: ")
    para i de 1 ate 5 faca
        escreval("Digite NOME[",i,"]:")
        leia(NOME[i])
        escreval("Digite PRECO[",i,"], em R$:")
        leia(PRECO[i])
        se (PRECO[i]<50) entao
            cont50 <- cont50 + 1
        fimse
    fimpara
    escreval("Existem ",cont50," produtos com preço menor que R$50")
    escreval("Produtos com preço entre R$50 e R$100: ")
    escreval
    para i de 1 ate 5 faca
        se (PRECO[i]>=50) e (PRECO[i]<=100) entao
            escreval(NOME[i])
        fimse
        se (PRECO[i] >= 100) entao
            somal00 <- somal00 + PRECO[i]
            cont100 <- cont100 + 1
        fimse
    fimpara
    medial00 <- somal00/cont100
    escreval("A média de preços de produtos com preço")
    escreval("superior a R$100,00 é de : R$ ",medial00:4:2)
finalgoritmo

```

4. Crie um algoritmo que leia um valor inteiro  $x$  e leia e carregue um vetor com 8 elementos inteiros e faça uma pesquisa no vetor, para saber em que posição encontramos o elemento  $x$  no vetor e, em caso contrário, exibir uma mensagem dizendo que o elemento  $x$  não aparece no vetor.

Ex:  $x = 666$

**v**

3	-8	666	50	666	23	-31	1
---	----	-----	----	-----	----	-----	---

Deve retornar a mensagem:  
666 está presente nas posições 3 e 5 do vetor

**Pesquisa ou Busca Sequencial em um Vetor**

3º.

1º.

2º.



```
algoritmo "Pesquisando no vetor"
var
    VET:vetor[1..8] de inteiro
    i,x,contx:inteiro
inicio
    contx <- 0
    //Rotina de leitura
    escreval("Digite o valor a ser pesquisado: ")
    leia(x)
    escreval("Carregando vetor VET:")
    para i de 1 ate 8 faca
        escreval("Digite VET[",i,"]:")
        leia(VET[i])
        se (VET[i] = x) entao
            contx <- contx + 1
        fimse
    fimpara
    se (contx > 0) entao
        escreval("Existem ",contx," números iguais a ",x," no vetor")
        escreval("nas posições: ")
        para i de 1 ate 8 faca
            se (VET[i]=x) entao
                escreval(i)
            fimse
        fimpara
    senao
        escreval("Não existem elementos iguais a ",x," no vetor")
    fimse
finalgoritmo
```

5. Faça um algoritmo que leia do teclado uma sequência de 10 números e utilize um vetor para armazená-los. Como saída, seu programa deve mostrar os números armazenados no vetor multiplicados por sua posição.


Exemplo:

Entrada									
1	2	3	4	5	6	7	8	9	10
1	2	3	4	5	6	7	8	9	10

Saída  
1 4 9 16 25 36 49 64 81 100

1º.

2º.



8* <b>1</b>	20*2	-6*3	3*4	11*5	-1*6	22*7	7* <b>8</b>	3*9	13*10
8	40	-18	12	55	-6	154	56	27	130

**Vetor Final (elementos \* índices)**

```
algoritmo "Vetor - multiplicação pelos seus índices"

var
    v: vetor[1..10] de inteiro
    i: inteiro
inicio
    escreval("Vetor - multiplicação pelos seus índices")
    //1o. passo - leitura do vetor de 10 inteiros
    escreval("Carregando o vetor de 10 inteiros: ")
    escreval("-----")
    para i de 1 ate 10 faca
        escreval("Digite v[",i,"]: ")
        leia(v[i])
    fimpara
    limpatela
    //2o. passo - impressão do vetor na tela
    escreval("Vetor de 10 elementos na tela: ")
    para i de 1 ate 10 faca
        escreva(v[i], " ")
    fimpara
    escreval("Vetor multiplicado pelos seus índices: ")
    para i de 1 ate 10 faca
        escreva(v[i] * i, " ")
    fimpara
finalgoritmo
```

Escopo	Nome	Tipo	Valor
GLOBAL	V[1]	I	5